VHDL 2008 - Use of custom floating package
See the formal for the generic package in float_generic_pkg.vhdl:

package fixed_pkg is new IEEE.fixed_generic_pkg
                           generic map (&lt;&gt;)


See IEEE Std 1076-2008 


  6.5.7.2 Generic map aspects  
  
  An actual associated with a formal generic package in a generic map aspect shall be a name that denotes an instance of the uninstantiated package named in the formal generic package declaration, as follows:
  
  
    ...
    b) If the formal generic package declaration includes an interface package generic map aspect in the form that includes the box (&lt;>) symbol, then the instantiated package denoted by the actual may be any instance of the uninstantiated package named in the formal generic package declaration.
  


You're missing an actual for the formal which is a generic package.

Instantiate IEEE.fixed_generic_pkg and provide the new instance name as the actual. 

architecture rtl of test is

use IEEE.fixed_float_types.all;
package fixed_pkg is new IEEE.fixed_generic_pkg
    generic map (
        fixed_round_style    =&gt; IEEE.fixed_float_types.fixed_round,
        fixed_overflow_style =&gt; IEEE.fixed_float_types.fixed_saturate,
        fixed_guard_bits     =&gt; 3,
        no_warning           =&gt; false
    );
package my_float_pkg is new ieee.float_generic_pkg
  generic map (
     float_exponent_width =&gt; 5,          -- 5 bits of exponent
     float_fraction_width =&gt; 11,         -- default will be float(5 dt –11)
     float_round_style    =&gt; round_zero, -- Truncate, don’t round
     float_denormalize    =&gt; false,      -- no denormal numbers
     float_guard_bits     =&gt; 0,          -- Unused by round_zero, set to 0
     float_check_error    =&gt; false,      -- Turn NAN and overflow off.
     no_warning           =&gt; true,       -- turn warnings off
     fixed_pkg            =&gt; fixed_pkg   -- ADDED
     );

begin

  process(data_A, data_B)
    variable data_A_float     : my_float_pkg.float(5 downto -11);
    variable data_B_float     : my_float_pkg.float(5 downto -11);
    variable data_out_float   : my_float_pkg.float(5 downto -11);
  begin

    data_A_float := my_float_pkg.to_float (
                    arg            =&gt; data_A(data_A_float'length - 1 downto 0), 
                    exponent_width =&gt; data_A_float'high, 
                    fraction_width =&gt; 0 - data_A_float'low  
                  );
  end process;
end architecture;


Also remove the prefix work from the references to my_float_pkg in the variable declarations in the process. Note the a prefix is added to the to_float call, my_float_pkg is not made visible by use clause.

That leaves you with an error for the to_float conversion function call which is outside the domain of this question. See the declaration in float_generic_pkg:

    -- std_ulogic_vector to float
       function to_float (
          arg                     : STD_ULOGIC_VECTOR;
          constant exponent_width : NATURAL := float_exponent_width; -- length of FP output exponent
          constant fraction_width : NATURAL := float_fraction_width) -- length of FP output fraction
         return UNRESOLVED_float;


The widths are given as naturals. The actual for the width of the fraction should be give the value 0 - data_A_float'low to make the number the natural length. (Note the references data_A'high and 'low were both adjusted to data_A_float'high and 'low, you were defining the wrong length as well).

And now the conversion works if the to_float call arg has the same length as the result type.

Note the IEEE fixed_pkg generic map for generic_fixed_pkg was used without regard to your actual usage (the possible impact of your generic constants of float_generic_pkg on my_float_pkg weren't evaluated). 

If you're going to use IEEE.fixed_pkg you can provide that as the actual instead of instantiating fixed_generic_pkg first.

In addition to not using using Synopsys package std_logic_unsigned (which has a -2008 IEEE equivalent numeric_std_unsigned) you also don't use package numeric_std.